THE UNIVERSITY OF SYDNEY

MATH2008 Introduction to Modern Algebra

(http://www.maths.usyd.edu.au/u/UG/IM/MATH2008/)

Semester 2, 2003 — Lecturer: R. Howlett

Computer Tutorial 12

This tutorial explores the groups of rotations of the Platonic solids. The Platonic solids are the five regular polytopes in three dimensions: the tetrahedron, the cube, the octahedron, the icosahedron and the dodecahedron. We shall represent the rotations of these solids as permutations of the vertices. In each case the full group of symmetries is twice as big as the group of rotations (and includes reflections and other kinds of transformations.)

1. The tetrahedron: Set up the group as

```
> T := PermutationGroup< 4 | (1,2,3), (1,3,4) >;
```

Check that T has order 12 and is equal to the alternating group $\mathrm{Alt}(4)$. Convince yourself (by looking at the diagram) that the elements of T give all possible rotations of the tetrahedron.

Solution.

```
> T:=PermutationGroup<4 | (1,2,3),(1,3,4)>;    > #T;    12
```

It is clear that $(1,2,3)$ and $(1,3,4)$ are rotational symmetries of the tetrahedron, and MAGMA tells us that the group T they generate has order 12. Since $(1,2,3)$ and $(1,3,4)$ are both even permutations, the group they generate must be a subgroup of the group of all even permutations, $\mathrm{Alt}(4)$. But $\mathrm{Alt}(4)$ has order 12 (since half the 24 permutations of $\{1,2,3,4\}$ are even) and MAGMA tells us that T has order 12; so $\mathrm{T} = \mathrm{Alt}(4)$. You can get MAGMA to confirm this: type `T eq Alt(4);` MAGMA will respond `true`.

```
> T eq Alt(4);
true
> Set(T);
{     (1, 2, 3),
      (1, 3, 4),
      (1, 2, 4),
      (1, 3)(2, 4),
      (1, 2)(3, 4),
      (2, 3, 4),
      (1, 3, 2),
      Id(T),
      (1, 4, 3),
      (1, 4)(2, 3),
      (1, 4, 2),
      (2, 4, 3)
}
```

For each vertex of the tetrahedron there are two rotational symmetries that fix that vertex: the line joining the vertex to the centroid of the opposite face is the axis of rotation, and you can rotate through either 120° or 240°. These rotations all have order 3, and correspond to the eight 3-cycles in $\mathrm{Alt}(4)$. For each edge of the tetrahedron there is a unique opposite edge (joining the two vertices that are not on the given edge). The rotation through 180° about the line joining the midpoints of a pair of opposite edges is a symmetry. This gives three more rotational symmetries. The identity is the 12th.

How do we know that there are no more rotational symmetries? Certainly the tetrahedron has some reflection symmetries (six, in fact). For example, the transposition $(1,2)$ corresponds to the reflection in the plane that is the perpendicular bisector of the edge joining vertices 1 and 2. (Note that vertices 3 and 4 lie in this plane.) Similarly, the other five transpositions in $\mathrm{Sym}(4)$ correspond to reflections in the planes that are the perpendicular bisectors of the other edges. Since every symmetry of the tetrahedron must correspond to some permutation of the vertices, the group of all symmetries must be some subgroup of $\mathrm{Sym}(4)$. So the order of the group of all symmetries must be a divisor of 24. Since we have already geometrically identified 18 symmetries, it follows that the symmetry group of the tetrahedron is the whole of $\mathrm{Sym}(4)$. The six symmetries that we have not yet geometrically identified correspond to the 4-cycles in $\mathrm{Sym}(4)$ (such as $(1,2,3,4)$). If ℓ is the line joining the midpoint of the edge 1-3 with the midpoint of the edge 2-4 then a rotation of 90° about the axis ℓ followed by the reflection in the plane that is the perpendicular bisector of ℓ is a symmetry of the tretrahedron corresponding to a 4-cycle. The other 4-cycles arise similarly.

It is not quite clear that these 4-cycles cannot also be described as rotations in some obscure way. To prove that they are definitely not rotations we need to use some linear algebra. Any rotation of $\mathbb{R}^3$ fixes all the points on some one-dimensional subspace ℓ (the axis of rotation). Let $\mathscr{P}$ be the plane through the origin perpendicular to the line ℓ. Then the rotation acts on $\mathscr{P}$ like a rotation of $\mathbb{R}^2$. If we now choose an orthonormal basis of $\mathbb{R}^3$ made up of one vector on ℓ and two in $\mathscr{P}$ then the matrix of the rotation has the form $\begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos\theta & -\sin\theta \\ 0 & \sin\theta & \cos\theta \end{pmatrix}$ (where θ is the angle of rotation). Since this matrix has determinant 1, we conclude that every rotation of $\mathbb{R}^3$ has determinant 1. A similar analysis can be used to show that reflections have determinant -1. And the transformations that correspond to the 4-cycles also have determinant -1, since they can each be described as the product of one reflection and one rotation. To sum all this up, the 12 even permutations in $\mathrm{Sym}(4)$ (i.e. the elements of $\mathrm{Alt}(4)$) correspond to rotational symmetries, and these all have determinant -1, while the 12 odd permutations in $\mathrm{Sym}(4)$ correspond to symmetries that have determinant -1.

It is also possible to use linear algebra to prove that the product of two rotations of $\mathbb{R}^3$ is also a rotation, and from this it follows readily that the set of all rotational symmetries of an object in $\mathbb{R}^3$ is always a group. Note that symmetries with determinant -1 cannot be physically performed on a rigid body in the ordinary space in which we live; so it is perhaps debatable whether or not they should be counted as “real” symmetries.





2. The cube: Set up the group as

```
> C:=PermutationGroup< 8 | (1,2,3,4)(8,7,6,5), (2,4,6)(7,5,3)>;
```

(*i*) Print the elements of C and use the diagram of the cube to work out the correspondence between rotations and permutations. Convince yourself that C contains all possible rotations of the cube.

(*ii*) As well as acting on the vertices of the cube the group acts on the four lines through opposite pairs of vertices. To see that the group just permutes these amongst themselves, type the following

```
> pairs := {1,8}^C;
> pairs;
```

To find the effect of the elements of C on these four pairs of vertices you can type the following:

```
> f,G,K := Action(C,pairs);
```

In carrying out this command MAGMA will construct a homomorphism f from C to the group of permutations of the set `pairs`. For each $g \in C$, $f(g)$ is the corresponding permutation. The group G is the image of f and the group K is its kernel.

(*iii*) Check that the image of f consists of all permutations of the four pairs and that the kernel contains only the identity element of C. It can be shown that a homomorphism whose kernel consists of the identity element only must be one-to-one. Conclude that the group of rotations of the cube is isomorphic to $\mathrm{Sym}(4)$.

(*iv*) If you look at the cube and think hard you should be able to see that there are three pairs of opposite faces and that the rotations of the cube permute these amongst themselves. In this part of the question you will construct a homomorphism from C to the group of permutations of these three pairs of faces. Here is the MAGMA code.

```
> faces := {{1,2,3,4},{5,6,7,8}}^C;
> print faces;
> f1,G1,K1 := Action(C,faces);
```

Check that the image of the homomorphism `f1` is the group of all permutations of the three pairs of faces, and conclude that it is isomorphic to $\mathrm{Sym}(3)$. Do you recognize the kernel?

Solution.

Given that the rotational symmetries of the cube form a group, it is easily seen that it has order 24. Imagine the cube placed on a desk. You can obviously rotate it so that any chosen face becomes the uppermost face. Since the faces are squares, there are then four possible rotations that leave the same face on top. This gives us all possible orientations. So the total number of rotations is the number of faces (six) times the number of rotations that fix a given face (four). This same argument applies to all the platonic solids: the order of the rotation group is the number of faces times the number of sides of each face.

```
> C:=PermutationGroup<8 | (1,2,3,4)(8,7,6,5), (2,4,6)(7,5,3)>;
> #C;
24
```



```
> Set(C);
{   (1, 2, 3, 4)(5, 8, 7, 6),
    (1, 7, 5)(2, 4, 8),
    (1, 3)(2, 4)(5, 7)(6, 8),
    (1, 3, 5)(4, 8, 6),
    (1, 4, 3, 2)(5, 6, 7, 8),
    (1, 2, 5, 6)(3, 8, 7, 4),
    (1, 6, 5, 2)(3, 4, 7, 8),
    (1, 8)(2, 7)(3, 4)(5, 6),
    (1, 4)(2, 7)(3, 6)(5, 8),
    (1, 2)(3, 6)(4, 5)(7, 8),
    (1, 5, 7)(2, 8, 4),
    (2, 6, 4)(3, 5, 7),
    (1, 5)(2, 6)(3, 7)(4, 8),
    (1, 7, 3)(2, 6, 8),
    (1, 8)(2, 3)(4, 5)(6, 7),
    (1, 5, 3)(4, 6, 8),
    (1, 6)(2, 7)(3, 8)(4, 5),
    Id(C),
    (1, 7)(2, 8)(3, 5)(4, 6),
    (1, 8)(2, 5)(3, 6)(4, 7),
    (1, 4, 7, 6)(2, 3, 8, 5),
    (1, 3, 7)(2, 8, 6),
    (2, 4, 6)(3, 7, 5),
}   (1, 6, 7, 4)(2, 5, 8, 3)
```

The six permutations here that are products of two 4-cycles correspond to 90° rotations (clockwise or anticlockwise) about axes joining the middle points of pairs of opposite faces. The 180° rotations about these axes give the permutations $(1,7)(4,6)(2,8)(3,5)$, $(1,3)(2,4)(5,7)(6,8)$ and $(1,5)(2,6)(3,7)(4,8)$. The other six permutations that are the products of four disjoint transpositions correspond to 180° rotations about axes that join midpoints of pairs of opposite edges. For each of the four pairs of opposite vertices there are two rotational symmetries of order 3: you can rotate clockwise or anticlockwise through 120° about the axis joining the opposite vertices. This gives the eight permutations in the above list that are products of two 3-cycles.

```
> pairs := {1,8}^C;
> pairs;
GSet{
    { 3, 6 },
    { 2, 7 },
    { 1, 8 },
}   { 4, 5 }
```

We chose the numbering of the vertices so that 8, 7, 6 and 5 are opposite to 1, 2, 3 and 4 respectively. Since it is clear that a symmetry that takes vertex i to vertex j must take the opposite of i to the opposite of j, every rotational symmetry of the cube gives rise to some permutation of the four pairs of opposite vertices. The MAGMA command `pairs := {1,8}^C` defined `pairs` to be the set of all pairs of the form $\{1^g, 8^g\}$, for g in the group `C`; the above output confirms that `pairs` $= \{\{1,8\},\{2,7\},\{3,6\},\{4,5\}\}$.

Let us write $\overline{1} = \{1,8\}$, $\overline{2} = \{2,7\}$, $\overline{3} = \{3,6\}$ and $\overline{4} = \{4,5\}$. It is easy to write down the permutations of $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$ corresponding to the 24 permutations in `C`. For example, $(1,2,3,4)(8,7,6,5)$ corresponds to $(\overline{1},\overline{2},\overline{3},\overline{4})$ and $(1,7,5)(2,4,8)$ corresponds to $(\overline{1},\overline{2},\overline{4})$. The MAGMA command `f,G,K:=Action(C,pairs)` defines `f` to be exactly this function from `C` to permutations of $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$ (but MAGMA has to write $\{1,8\}$ rather than $\overline{1}$, etc.).

```
> f,G,K := Action(C,pairs);
> f(C!(1,2,3,4)(8,7,6,5));
({ 3, 6 }, { 4, 5 }, { 1, 8 }, { 2, 7 })
> f(C!(1,7,5)(2,4,8));
({ 2, 7 }, { 4, 5 }, { 1, 8 })
```

The group `G`, the image of `f`, consists of all 24 permutations of $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$. That is, every permutation of $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$ arises as `f(`c`)` for some c in `C`.

```
> G;
Permutation group G acting on a set of cardinality 4
    ({ 3, 6 }, { 4, 5 }, { 1, 8 }, { 2, 7 })
    ({ 3, 6 }, { 2, 7 }, { 4, 5 })
> #G;
24
> K;
Permutation group K acting on a set of cardinality 8
Order = 1
```

Since `G` has the same number of elements as `C` it follows that the mapping `f` must be one-to-one. (In particular, there is only one element of `C` that gets mapped to the identity permutation of `pairs`. That is, the kernel `K` has only one element.) So `C` is isomorphic to `G`; that is, `C` is essentially just Sym(4).

```
> faces := {{1,2,3,4},{5,6,7,8}}^C;
> faces;
GSet{
    {
        { 1, 2, 3, 4 },
        { 5, 6, 7, 8 }
    },
    {
        { 1, 2, 5, 6 },
        { 3, 4, 7, 8 }
    },
    {
        { 1, 4, 6, 7 },
        { 2, 3, 5, 8 }
    }
}
```



```
> f1,G1,K1 := Action(C,faces);
> #G1;
6
> K1;
Permutation group K1 acting on a set of cardinality 8
Order = 4 = 2^2
    (1, 3)(2, 4)(5, 7)(6, 8)
    (1, 7)(2, 8)(3, 5)(4, 6)
> f(K1);
Permutation group acting on a set of cardinality 4
    ({ 3, 6 }, { 1, 8 })({ 2, 7 }, { 4, 5 })
    ({ 3, 6 }, { 4, 5 })({ 2, 7 }, { 1, 8 })
> Set(f(K1));
{
    Id($),
    ({ 3, 6 }, { 1, 8 })({ 2, 7 }, { 4, 5 }),
    ({ 3, 6 }, { 2, 7 })({ 1, 8 }, { 4, 5 }),
    ({ 3, 6 }, { 4, 5 })({ 2, 7 }, { 1, 8 })
}
```

There are three pairs of opposite faces, and the group `G1` is a group of permutations of these three objects. Since `G1` has 6 elements, it must consist of all permutations of the pairs of opposite faces. Since `f1` takes the 24 elements of `C` to the 6 elements of `G1`, the map `f1` is certainly not one-to-one. In fact, `C` is essentially Sym(4) and `G1` is essentially Sym(3), and `f1` is the same homomorphism from permutations of 4 things to permutations of three things that we described in lectures (see Question 2 of Computer Tutorial 10). The kernel of this homomorphism consists of the identity and the three permutations in Sym(4) that are products of of two disjoint transpositions. The MAGMA output above confirms this: `Set(f(K1))` lists the permutations of $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$ corresponding to elements of `K1`, and we see that they are id, $(\overline{3},\overline{1})(\overline{2},\overline{4})$, $(\overline{3},\overline{2})(\overline{1},\overline{4})$, $(\overline{3},\overline{4})(\overline{2},\overline{1})$.

3. The group of the tetrahedron is isomorphic to Alt(4) and the group of the cube is isomorphic to Sym(4). In fact it is possible to place two tetrahedra inside the cube in such a way that the even permutations in Sym(4) fix the tetrahedra setwise and the odd permutations in Sym(4) interchange the two tetrahedra. The two tetrahedra are $t_1 = \{2,4,6,8\}$ and $t_2 = \{1,3,5,7\}$.

(i) Check that every element of C either leaves t_1 in place or sends it to t_2. Do this with the following MAGMA code.

```
> t1 := {2,4,6,8};
> for g in C do
for> t1^g;
for> end for;
```

(ii) Use MAGMA to find the stabilizer of t_1. That is,
`> H := Stabilizer(C,t1);`What is the order of H? Is this one of the groups you have seen before? Which one?

Solution.

Two vertices of the cube are said to be *adjacent* if they are connected by an edge. It is possible to colour the vertices black and white in such a way that adjacent vertices are always oppositely coloured. Four of the eight vertices will be white and four black; going diagonally across faces moves you between vertices of the same colour.

Since symmetries take adjacent vertices to adjacent vertices, a symmetry that takes a black vertex A to a white vertex B will take the white vertices adjacent to A to the black vertices adjacent to B. It follows readily that such a symmetry must take all the black vertices to white vertices and all the white vertices to black vertices. And similarly a symmetry that takes one black vertex to a black vertex must take all the black vertices to black vertices. Our numbering of the vertices is such that the odd-numbered vertices are one colour and the even-numbered vertices the opposite colour.

```
> t1 := {2,4,6,8};
> for g in C do
for> t1^g;
for> end for;
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
```

```
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
{ 2, 4, 6, 8 }
{ 1, 3, 5, 7 }
>  H := Stabilizer(C,t1);
> #H;
12
> H;
Permutation group H acting
          on a set of cardinality 8
Order = 12 = 2^2 * 3
    (1, 3, 7)(2, 8, 6)
    (2, 6, 4)(3, 5, 7)
> f(H);
Permutation group acting on
             a set of cardinality 4
    ({ 3, 6 }, { 2, 7 }, { 1, 8 })
    ({ 3, 6 }, { 4, 5 }, { 2, 7 })
```

The black vertices of the cube can be regarded as the vertices of a tetrahedron, the edges of which run diagonally across the faces of the cube. The rotations of the cube that take black vertices to black vertices are symmetries of this tetrahedron. Of course, these same symmetries also take white vertices to white vertices, and are thus also symmetries of the tetrahedron made up by the white vertices.

In the MAGMA output above, `H` is the group of colour-preserving rotations of `C`. Half the 24 elements of `C` are in `H`. We know that `C` is isomorphic to Sym(4), and `H` is isomorphic to Alt(4). The output above confirms that the isomorphism `f` takes `H` to the group generated by the 3-cycles $(\overline{3},\overline{2},\overline{1})$ and $(\overline{3},\overline{4},\overline{2})$, and (cf. Q1)

this is the alternating group on $\{\overline{1},\overline{2},\overline{3},\overline{4}\}$.



4. The Octahedron:

```
> O:=PermutationGroup< 6 | (1,2,3)(4,6,5), (2,3,5,4)>;
```

(*i*) Print out the elements of *O* and convince yourself that all rotations of the octahedron are accounted for.

(*ii*) What is the order of *O*?

(*iii*) Observe that if you put a vertex at the centre of each face, these eight new vertices will describe a cube. Thus the group *O* must be isomorphic to *C*, the group of rotations of the cube. Use MAGMA to construct the action of *O* on its faces:

```
> triples := {1,2,3}^O;
> print triples;
> g, H, L := Action(O,triples);
```

Solution.

```
> O:=PermutationGroup
    <6|(1,2,3)(4,6,5),(2,3,5,4)>;
> Set(O);
{
    (1, 6)(2, 3)(4, 5),
    (1, 4, 2)(3, 5, 6),
    (1, 5)(2, 6)(3, 4),
    (1, 3, 2)(4, 5, 6),
    (1, 2, 6, 5),
    (1, 5, 3)(2, 4, 6),
    (1, 3)(2, 5)(4, 6),
    (1, 6)(2, 4)(3, 5),
    (1, 2)(3, 4)(5, 6),
    (1, 4, 5)(2, 6, 3),
    (2, 4, 5, 3),
```

```
    (1, 6)(2, 5),
    Id(O),
    (2, 5)(3, 4),
    (2, 3, 5, 4),
    (1, 2, 4)(3, 6, 5),
    (1, 3, 6, 4),
    (1, 4)(2, 5)(3, 6),
    (1, 5, 4)(2, 3, 6),
    (1, 4, 6, 3),
    (1, 5, 6, 2),
    (1, 3, 5)(2, 6, 4),
    (1, 2, 3)(4, 6, 5),
    (1, 6)(3, 4)
}
```

There are four pairs of opposite faces; rotations of 120° and 240° about the axes joining centroids of opposite faces are symmetries of the octahedron. This gives the 8 elements of order 3 in `O`. There are three pairs of opposite vertices, giving three axes for which there are rotations through 90° and 270°. This gives the six rotations of order 4. Rotations through 180° about these same three axes are also symmetries: these correspond to the permutations that are products of two disjoint 2-cycles. There are 6 pairs of opposite edges, and for each pair there is a rotation through 180° about the axis through the midpoints of the two edges. Together with the identity, this makes 24 rotations. Since the octahedron has 8 triangular faces, this is in agreement with the formula stated in the solution to Q2 above: the order of the group is the product of the number of faces and the number of edges of each face. As with the group `C` there are 8 elements of order 3, 6 of order 4 and 9 of order 2, three of these 9 being the squares of the elements of order 4. As explained in the question, `O` is in fact isomorphic to `C`.

```
> #O;
24
> triples := {1,2,3}^O;
> print triples;
GSet{
    { 3, 5, 6 },
    { 1, 3, 5 },
    { 2, 3, 6 },
    { 1, 4, 5 },
    { 1, 2, 3 },
    { 2, 4, 6 },
    { 1, 2, 4 },
    { 4, 5, 6 }
}
> g, H, L := Action(O,triples);
> H;
Permutation group H acting on a set of cardinality 8
 ({3,5,6},{1,4,5},{2,4,6})({1,3,5},{1,2,4},{2,3,6})
 ({3,5,6},{4,5,6},{2,4,6},{2,3,6})({1,3,5},{1,4,5},
                                      {1,2,4},{1,2,3})
> #H;
24
> L;
Permutation group L acting on a set of cardinality 6
Order = 1
```

The set $\{1,2,3\}$ corresponds to one of the faces of the octahedron, and `triples` is defined to be the set of everything obtained from $\{1,2,3\}$ by a rotation of the octahedron. So `triples` is the set of all faces of the octahedron. It is easy to use the diagram to check that MAGMA's list of the elements of `triples` is correct.

The above MAGMA code defines `g` to be a homomorphism from `O` onto `H`, which is a group of permutations of `triples`, the set of faces the octahedron. The homomorphism is an isomorphism since `H` has the same number of elements as `O`. This also means that the kernel `L` must have order 1.

We can associate the faces of the octahedron with the vertices of the cube as follows: $\{1,2,3\} \leftrightarrow 1$, $\{1,3,5\} \leftrightarrow 2$, $\{1,5,4\} \leftrightarrow 3$, $\{1,4,2\} \leftrightarrow 4$, $\{6,5,4\} \leftrightarrow 8$, $\{6,4,2\} \leftrightarrow 7$, $\{6,2,3\} \leftrightarrow 6$, $\{6,3,5\} \leftrightarrow 5$. The generators MAGMA gave for `H` above correspond exactly to the generators we gave for `C`, showing that `H` and `C` are isomorphic.




5. The Dodecahedron and Icosahedron:

```
> I:=PermutationGroup<12 |
    (2,3,4,5,6)(11,10,9,8,7), (1,6,9,8,3)(12,7,4,5,10)>;
> D:=PermutationGroup<20 |
    (1,2,3,4,5)(10,9,8,7,6)(11,12,13,14,15)(20,19,18,17,16),
    (2,5,9)(3,10,15)(4,14,8)(6,18,11)(7,13,17)(12,19,16)>;
```

(*i*) The icosahedron and the dodecahedron are *dual* in that you can get one from the other by putting a vertex in the centre of each face. Therefore they have the same group of rotations. Check the order of each group to get a first indication of this. What other groups of this order have you met in this course?

(*ii*) Not all the vertices are visible in the diagrams. Can you work out the numbering of the hidden faces? (Hint. Use the group to find the images of one of the faces you can see, following the steps of the previous question.)

(*iii*) The dodecahedron has 20 vertices and it turns out that you can divide these up into five lots of four so that each set of four is a tetrahedron. There are two ways to do this. One of them is given as follows:

```
> tetra := {1,6,11,18}^D;
> tetra;
```

You should find that there are five sets in `tetra`. This means that D permutes these 5 tetrahedra. In fact D is the alternating group of this set of 5 tetrahedra. Can you explain why this is so?

Solution.

```
> I:=PermutationGroup<12 | (2,3,4,5,6)(11,10,9,8,7),
(1,6,9,8,3)(12,7,4,5,10)>;
> D:=PermutationGroup<20 |
    (1,2,3,4,5)(10,9,8,7,6)(11,12,13,14,15)(20,19,18,17,16),
       (2,5,9)(3,10,15)(4,14,8)(6,18,11)(7,13,17)(12,19,16)>;
> #D;
60
> #I;
60
```

Our numbering of the 20 vertices of the dodecahedron can be described as follows. On one of the faces the vertices are numbered 1,2,3,4,5 in anticlockwise order. There is another face containing the edge joining vertices 1 and 2. The vertices on this face are 2,1,9,15,8 (also in anticlockwise order). The full list of faces (all given in anticlockwise order) is as follows:

$1,2,3,4,5$	$2,1,9,15,8$	$3,2,8,11,7$	$4,3,7,12,6$
$5,4,6,13,10$	$1,5,10,14,9$	$15,9,14,18,17$	$11,8,15,17,16$
$12,7,11,16,20$	$13,6,12,20,19$	$14,10,13,19,18$	$16,17,18,19,20$

Numbers on opposite vertices add up to 21.

The 12 vertices of the icosahedron are numbered in accordance with the following description. The five vertices adjacent to vertex 1 are 2,3,4,5,6 (anticlockwise). Opposite vertices add up to 13; so the vertices adjacent to vertex 12 are 11,10,9,8,7. The faces are as follows:

$1,2,3$	$1,3,4$	$1,4,5$	$1,5,6$	$1,6,2$
$2,8,3$	$3,7,4$	$4,11,5$	$5,10,6$	$6,9,2$
$2,9,8$	$3,8,7$	$4,7,11$	$5,11,10$	$6,10,9$
$12,7,8$	$12,8,9$	$12,9,10$	$12,10,11$	$12,11,7$

The groups `I` and `D` must be isomorphic since the icosahedron and the dodecahedron are dual to one another. They have 60 elements. The alternating group Alt(5) also has 60 elements, and in fact `I` and `D` are isomorphic to Alt(5).

```
> tetra := {1,6,11,18}^D;
> tetra;
GSet{
    { 2, 10, 12, 17 },
    { 4, 8, 14, 20 },
    { 5, 7, 15, 19 },
    { 3, 9, 13, 16 },
    { 1, 6, 11, 18 }
}
> k,M,N := Action(D,tetra);
> k;
Mapping from: GrpPerm: D to GrpPerm: M
> M;
Permutation group M acting on a set of cardinality 5
  ({2,10,12,17},{3,9,13,16},{4,8,14,20},{5,7,15,19},{1,6,11,18})
  ({2,10,12,17},{5,7,15,19},{3,9,13,16})
> #M;
60
> N;
Permutation group N acting on a set of cardinality 20
Order = 1
```

The homomorphism `k` constructed here takes `D` onto `M`, a group of permutations of the five tetrahedra. Since `M` has the same number of elements as `D` they must be isomorphic.

In order to properly understand this isomorphism, it is necessary to study a real dodecahedron, not just a picture. The five tetrahedra can be described as follows.

Imagine an insect that crawls along the edges of the dodecahedron. Since three edges meet at each vertex, when the insect reaches a vertex it will have a choice of two edges to crawl along next (assuming it doesn't turn round): it can take the left fork or the right fork. Suppose now that the insect starts at vertex A, crawls along an edge to an adjacent vertex, turns to the right and crawls to the next vertex, then turns to the left, crawls along one more edge and then stops. Call the vertex it reaches like this vertex B. Now suppose the insect returns to vertex A and repeats the process, but chooses a different edge leading from A to start with. Call the vertex it reaches C. And starting again from A, take the third edge leading from A and repeat the procedure (turn right at the first fork, left at the second) to reach vertex D. Then A, B, C and D are the vertices of a regular tetrahedron. Any symmetry of the dodecahedron will clearly take a tetrahedron found by the above procedure to another one. Since it turns out that there are precisely five such tetrahedra, the group `D` permutes these five objects. Using the classification of rotational symmetries of the regular tetrahedron (see Q1) it is not hard to see that there is no rotation of $\mathbb{R}^3$ apart from the identity that preserves all five of these tetrahedra. Since `D` has order 60, it follows that we obtain 60 distinct permutations of the five tetrahedra. It can be shown that Alt(5) is the only subgroup of Sym(5) with 60 elements; so `D` must be isomorphic to Alt(5). Alternatively, one can check directly that the generators of `D` give rise to even permutations of `tetra`. The MAGMA function `Sign` returns $+1$ for even permutations and -1 for odd permutations.

```
> x:=D!(1,2,3,4,5)(10,9,8,7,6)(11,12,13,14,15)(20,19,18,17,16);
> y:=D!(2,5,9)(3,10,15)(4,14,8)(6,18,11)(7,13,17)(12,19,16);
> k(x);
({2,10,12,17},{3,9,13,16},{4,8,14,20},{5,7,15,19},{1,6,11,18})
> Sign(k(x));
1
> k(y);
({2,10,12,17},{5,7,15,19},{3,9,13,16})
> Sign(k(y));
1
```

The elements `x` and `y` generate `D`. The isomorphism `k` takes them to permutations of `tetra` that are (repectively) a 5-cycle and a 3-cycle. Since 5-cycles and 3-cycles are both even, and products of even permutations are even, all things in the group generated by `x` and `y` will also give rise to even permutations of `tetra`. So `D` is isomorphic to a group of even permutations of the set 5-element set `tetra`, and since $\#D = 60$, it is isomorphic to the whole of Alt(5).